# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Daniel Wallace <danielwallace at gtmanfred dot com>

pkgname=python-openstackclient
pkgver=6.2.0
_commit=05d34ffce9e710021b0d125add38f2fd73c7852d
pkgrel=2
pkgdesc="OpenStack Command-line Client"
arch=('any')
url="https://docs.openstack.org/python-openstackclient/latest/"
license=('Apache')
depends=('python-six' 'python-pbr' 'python-babel' 'python-cliff' 'python-keystoneclient'
         'python-keystoneauth1' 'python-osc-lib' 'python-oslo-utils' 'python-oslo-i18n'
         'python-openstacksdk' 'python-glanceclient' 'python-novaclient' 'python-cinderclient')
makedepends=('git')
checkdepends=('python-ddt' 'python-oslotest' 'python-requests-mock')
source=("git+https://github.com/openstack/python-openstackclient.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd python-openstackclient
  python setup.py build
}

check() {
  cd python-openstackclient
  stestr run
}

package() {
  cd python-openstackclient
  python setup.py install --root="$pkgdir" --optimize=1
}
